


Problem-2.20. Bant Matrisi

$$A = \begin{bmatrix} a_{11} & a_{12} & 0 & 0 \\ a_{21} & a_{22} & a_{23} & 0 \\ 0 & a_{32} & a_{33} & a_{34} \\ 0 & 0 & a_{34} & a_{44} \end{bmatrix}$$
 şeklinde verilen bir A matrisini ekrana yazdıracak programı yazalım.

$n \times n$ boyutlu $0 \leq k \leq n-1$ tamsayısı için $|i-j| > k$ iken $a_{ij} = 0$ olarak tanımlanan matrislere $2k+1$ genişlikli bant matrisi denir. Bir bant matrisde, tüm sıfır olmayan elemanlar ana köşegen üzerinde ve ana köşegenin üstünde ve altındaki ilk k köşegende yerleşmiştir. Böyle bir matrisin genel biçimi bir bant gibidir ve



şekindedir. Bir köşegen matris, $k=0$ ile bir bant matristir. Buna göre verilen k ve n değerlerine göre istenilen bant matrisi ekrana yazdıracak program aşağıdaki gibi yazılabilir.

```
INPUT "Karesel matrisin boyutunu giriniz=", n
INPUT "K sayısını giriniz=", k
DIM A(n, n)
FOR i = 1 TO n
  FOR j = 1 TO n
    IF ABS(i - j) <= k THEN
      PRINT "A("; i; ", "; j; ")="
      INPUT A(i, j)
    ELSE
      'A(i, j) = 0
    END IF
  NEXT j
NEXT i
FOR i = 1 TO n
  FOR j = 1 TO n
    PRINT A(i, j);
  NEXT j
  PRINT
NEXT i
```

Problem-2.21. Köşegen ve Ters Köşegen Elemanları 1 Diğer Elemanları 0 Olan Matris

$$A = \begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 1 & 0 \\ 0 & 1 & 1 & 0 \\ 1 & 0 & 0 & 1 \end{bmatrix}$$

şeklinde verilen A matrisini ekrana yazdıracak programı yazalım. Matrise dikkat edilirse köşegen ve ters köşegen elemanları 1 diğer elemanları 0 olan bir matristir. Bu şekilde verilen bir matrisi ekrana yazdıracak program aşağıdaki gibi yazılabilir.

```
INPUT "Karesel matrisin boyutunu giriniz=", n
DIM A(n, n)
FOR i = 1 TO n
  A(i, i) = 1
  A(i, n + 1 - i) = 1
NEXT i
FOR i = 1 TO n
  FOR j = 1 TO n
    PRINT A(i, j);
  NEXT j
  PRINT
NEXT i
```

Problem-2.22. n tek sayı olmak üzere $(n+1)/2$ nci Satır ve Sütun Elemanları 1, Diğerleri 0 Olan Matris

n tek bir sayı olmak üzere $n \times n$ boyutlu,

$$A = \begin{bmatrix} 0 & 0 & \dots & \dots & \dots & 0 & 1 & 0 & \dots & \dots & \dots & 0 & 0 \\ 0 & 0 & \dots & \dots & \dots & 0 & 1 & 0 & \dots & \dots & \dots & 0 & 0 \\ \dots & \dots & \dots & \dots & \dots & \dots & 1 & \dots & \dots & \dots & \dots & \dots & \dots \\ \dots & \dots & \dots & \dots & \dots & \dots & 1 & \dots & \dots & \dots & \dots & \dots & \dots \\ \dots & \dots & \dots & \dots & \dots & \dots & 1 & \dots & \dots & \dots & \dots & \dots & \dots \\ 1 & 1 & 1 & 1 & 1 & 1 & 1 & 1 & 1 & 1 & 1 & 1 & 1 \\ \dots & \dots & \dots & \dots & \dots & \dots & 1 & \dots & \dots & \dots & \dots & \dots & \dots \\ \dots & \dots & \dots & \dots & \dots & \dots & 1 & \dots & \dots & \dots & \dots & \dots & \dots \\ \dots & \dots & \dots & \dots & \dots & \dots & 1 & \dots & \dots & \dots & \dots & \dots & \dots \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

↓
 $(n+1)/2$

$(n+1)/2$

matrisini ekrana yazdıracak programı yazalım. Karesel matrisin boyutu (n) tek olmak üzere $(n+1)/2$ nci satır ve sütunu 1 ve diğer elemanları 0 olan matrisi ekrana yazdıracak program aşağıdaki gibi yazılabilir.

```
INPUT "Tek sayı olacak biçimde karesel matrisin boyutunu giriniz=", n
DIM A(n, n)
k = (n + 1) / 2
FOR i = 1 TO n
    A(i, k) = 1
    A(k, i) = 1
NEXT i
FOR i = 1 TO n
    FOR j = 1 TO n
        PRINT A(i, j);
    NEXT j
    PRINT
NEXT i
```

Problem-2.23. Matrisin Elemanlarını Diziye, Dizinin Elemanlarını Matrise Aktarma

Verilen $n \times m$ boyutlu bir matrisin elemanlarının diziye aktarılması, oluşturulan dizinin elemanlarının yeni bir matrise aktarılması ve sonuçların ekrana yazdırılması problemi için programı yazalım. Burada dikkat edilmesi gereken nokta matris elemanlarının aktarılacağı dizinin boyutunun uygun bir biçimde belirlenmesidir. n satır, m sütundan oluşan matrisin $n \times m$ tane elemanı olacağından dizinin boyutu da $n \times m$ kadar olur.

Örneğin, $n=2$, $m=3$ için matrisin elemanlarının diziye aktarılma işlemi,

```
dizi(1)=matris(1, 1)
dizi(2)=matris(1, 2)
dizi(3)=matris(1, 3)
dizi(4)=matris(2, 1)
dizi(5)=matris(2, 2)
dizi(6)=matris(2, 3)
```

biçiminde yapılır.

```
INPUT "SATIR SAYISINI GİRİNİZ"; n
INPUT "SÜTUN SAYISINI GİRİNİZ"; m
DIM matris(n, m): DIM DIZI(n * m): DIM A(n, m)
FOR I = 1 TO n
    FOR J = 1 TO m
        PRINT "matris("; I; ", "; J; ")";
        INPUT matris(I, J)
```

```

NEXT J, I
FOR I = 1 TO n
  FOR J = 1 TO m
    PRINT matris(I, J);
  NEXT J
  PRINT
NEXT I
REM*** diziye aktarma işlemi
FOR I = 1 TO n
  FOR J = 1 TO m
    T = T + 1
    DIZI(T) = matris(I, J)
  NEXT J
NEXT I
REM*** dizinin elemanlarını yazdırıyoruz***
PRINT "OLUŞTURULAN DİZİ"
FOR I = 1 TO n * m
  PRINT DIZI(I)
NEXT I
PRINT
PRINT
REM*** diziyi matrise aktarıyoruz***
FOR I = 1 TO n
  FOR J = 1 TO m
    L = L + 1
    A(I, J) = DIZI(L)
  NEXT J, I
REM*** dizinin aktarıldığı matrisin elemanlarını yazdırıyoruz
PRINT "Oluşturulan Matris"
FOR I = 1 TO n
  FOR J = 1 TO m
    PRINT USING "###.#"; A(I, J);
  NEXT J
  PRINT
NEXT I

```